<commit_before>DESCRIPTION = "GPM (General Purpose Mouse) is a mouse server \
for the console and xterm, with sample clients included \
(emacs, etc)."
SECTION = "console/utils"
LICENSE = "GPLv2+"
LIC_FILES_CHKSUM = "file://COPYING;md5=18810669f13b87348459e611d31ab760"

PV = "1.99.7+git${SRCREV}"
PR = "r2"
SRCREV = "1fd19417b8a4dd9945347e98dfa97e4cfd798d77"

DEPENDS = "ncurses bison-native"

SRC_URI = "git://github.com/telmich/gpm;protocol=git \
           file://init \
           file://no-docs.patch \
           file://processcreds.patch \
           file://gpm.service.in \
           file://0001-Use-sigemptyset-API-instead-of-__sigemptyset.patch \
           "

S = "${WORKDIR}/git"

inherit autotools-brokensep update-rc.d systemd

INITSCRIPT_NAME = "gpm"
INITSCRIPT_PARAMS = "defaults"

do_configure_prepend() {
    (cd ${S};./autogen.sh;cd -)
}

do_install_append () {
    if ${<EMAIL>('DISTRO_FEATURES', 'systemd', 'true', 'false', d)}; then
        install -d ${D}${systemd_system_unitdir}
        sed 's:@bindir@:${bindir}:' < ${WORKDIR}/gpm.service.in >${D}${systemd_system_unitdir}/gpm.service
    fi
    if ${<EMAIL>('DISTRO_FEATURES', 'sysvinit', 'true', 'false', d)}; then
        install -D -m 0755 ${WORKDIR}/init ${D}/${sysconfdir}/init.d/gpm
    fi
    install -D -m 0644 ${S}/src/headers/gpm.h ${D}${includedir}/gpm.h
    ln -s libgpm.so.2 ${D}${libdir}/libgpm.so
}

SYSTEMD_SERVICE_${PN} = "gpm.service"

FILES_${PN} += "${datadir}/emacs"<commit_msg>gpm: Fix gpm path in unit file<commit_after>DESCRIPTION = "GPM (General Purpose Mouse) is a mouse server \
for the console and xterm, with sample clients included \
(emacs, etc)."
SECTION = "console/utils"
LICENSE = "GPLv2+"
LIC_FILES_CHKSUM = "file://COPYING;md5=18810669f13b87348459e611d31ab760"

PV = "1.99.7+git${SRCREV}"
PR = "r2"
SRCREV = "1fd19417b8a4dd9945347e98dfa97e4cfd798d77"

DEPENDS = "ncurses bison-native"

SRC_URI = "git://github.com/telmich/gpm;protocol=git \
           file://init \
           file://no-docs.patch \
           file://processcreds.patch \
           file://gpm.service.in \
           file://0001-Use-sigemptyset-API-instead-of-__sigemptyset.patch \
           "

S = "${WORKDIR}/git"

inherit autotools-brokensep update-rc.d systemd

INITSCRIPT_NAME = "gpm"
INITSCRIPT_PARAMS = "defaults"

do_configure_prepend() {
    (cd ${S};./autogen.sh;cd -)
}

do_install_append () {
    if ${<EMAIL>('DISTRO_FEATURES', 'systemd', 'true', 'false', d)}; then
        install -d ${D}${systemd_system_unitdir}
        sed 's:@bindir@:${sbindir}:' < ${WORKDIR}/gpm.service.in >${D}${systemd_system_unitdir}/gpm.service
    fi
    if ${<EMAIL>('DISTRO_FEATURES', 'sysvinit', 'true', 'false', d)}; then
        install -D -m 0755 ${WORKDIR}/init ${D}/${sysconfdir}/init.d/gpm
    fi
    install -D -m 0644 ${S}/src/headers/gpm.h ${D}${includedir}/gpm.h
    ln -s libgpm.so.2 ${D}${libdir}/libgpm.so
}

SYSTEMD_SERVICE_${PN} = "gpm.service"

FILES_${PN} += "${datadir}/emacs"
{'fraction_non_alphanumeric': 0.17799461641991923, 'fraction_numerical': 0.0477792732166891, 'mean_word_length': 3.4044444444444446, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 9, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6121525', 'n_tokens_mistral': 1289, 'n_tokens_neox': 1186, 'n_words': 203}